NAG Library Routine Document

F11DEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F11DEF solves a real sparse nonsymmetric system of linear equations, represented in co-ordinate storage format, using a restarted generalized minimal residual (RGMRES), conjugate gradient squared (CGS), stabilized bi-conjugate gradient (Bi-CGSTAB), or transpose-free quasi-minimal residual (TFQMR) method, without preconditioning, with Jacobi, or with SSOR preconditioning.

2 Specification

```
      SUBROUTINE F11DEF(METHOD, PRECON, N, NNZ, A, IROW, ICOL, OMEGA, B, M,
     1                  TOL, MAXITN, X, RNORM, ITN, WORK, LWORK, IWORK, IFAIL)
      INTEGER           N, NNZ, IROW(NNZ), ICOL(NNZ), M, MAXITN, ITN, LWORK,
     1                  IWORK(2*N+1), IFAIL
      double precision  A(NNZ), OMEGA, B(N), TOL, X(N), RNORM, WORK(LWORK)
      CHARACTER*(*)     METHOD
      CHARACTER*1       PRECON
```

3 Description

F11DEF solves a real sparse nonsymmetric system of linear equations

$$Ax = b,$$

using an RGMRES (see Saad and Schultz (1986)), CGS (see Sonneveld (1989)), Bi-CGSTAB(ℓ) (see Van der Vorst (1989) and Sleijpen and Fokkema (1993)), or TFQMR (see Freund and Nachtigal (1991) and Freund (1993)) method.

The routine allows the following choices for the preconditioner:

no preconditioning;

Jacobi preconditioning (see Young (1971));

symmetric successive-over-relaxation (SSOR) preconditioning (see Young (1971)).

For incomplete LU (ILU) preconditioning see F11DCF.

The matrix A is represented in co-ordinate storage (CS) format (see Section 2.1.1 in the F11 Chapter Introduction) in the arrays A, IROW and ICOL. The array A holds the nonzero entries in the matrix, while IROW and ICOL hold the corresponding row and column indices.

F11DEF is a Black Box routine which calls F11BDF, F11BEF and F11BFF. If you wish to use an alternative storage scheme, preconditioner, or termination criterion, or require additional diagnostic information, you should call these underlying routines directly.

4 References

Freund R W (1993) A transpose-free quasi-minimal residual algorithm for non-Hermitian linear systems *SIAM J. Sci. Comput.* **14** 470–482

Freund R W and Nachtigal N (1991) QMR: a Quasi-Minimal Residual Method for Non-Hermitian Linear Systems *Numer. Math.* **60** 315–339

Saad Y and Schultz M (1986) GMRES: a generalized minimal residual algorithm for solving nonsymmetric linear systems *SIAM J. Sci. Statist. Comput.* **7** 856–869

Sleijpen G L G and Fokkema D R (1993) BiCGSTAB(ℓ) for linear equations involving matrices with complex spectrum *ETNA* **1** 11–32

Sonneveld P (1989) CGS, a fast Lanczos-type solver for nonsymmetric linear systems *SIAM J. Sci. Statist. Comput.* **10** 36–52

Van der Vorst H (1989) Bi-CGSTAB, a fast and smoothly converging variant of Bi-CG for the solution of nonsymmetric linear systems *SIAM J. Sci. Statist. Comput.* **13** 631–644

Young D (1971) *Iterative Solution of Large Linear Systems* Academic Press, New York

5 Parameters

1: METHOD – CHARACTER*(*) *Input*

On entry: the iterative method to be used.

METHOD = 'RGMRES'
Restarted generalized minimum residual method.

METHOD = 'CGS'
Conjugate gradient squared method.

METHOD = 'BICGSTAB'
Bi-conjugate gradient stabilized (ℓ) method.

METHOD = 'TFQMR'
Transpose-free quasi-minimal residual method.

Constraint: METHOD = 'RGMRES', 'CGS' or 'BICGSTAB', or 'TFQMR'.

2: PRECON – CHARACTER*1 *Input*

On entry: specifies the type of preconditioning to be used.

PRECON = 'N'
No preconditioning.

PRECON = 'J'
Jacobi.

PRECON = 'S'
Symmetric successive-over-relaxation.

Constraint: PRECON = 'N', 'J' or 'S'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 1.

4: NNZ – INTEGER *Input*

On entry: the number of nonzero elements in the matrix A.

Constraint: $1 \leq \text{NNZ} \leq \text{N}^2$.

5: A(NNZ) – ***double precision*** array *Input*

On entry: the nonzero elements of the matrix A, ordered by increasing row index, and by increasing column index within each row. Multiple entries for the same row and column indices are not permitted. The routine F11ZAF may be used to order the elements in this way.

6: IROW(NNZ) – INTEGER array *Input*
7: ICOL(NNZ) – INTEGER array *Input*

On entry: the row and column indices of the nonzero elements supplied in A.

Constraints:

IROW and ICOL must satisfy the following constraints (which may be imposed by a call to F11ZAF):

$1 \leq \text{IROW}(i) \leq \text{N}$ and $1 \leq \text{ICOL}(i) \leq \text{N}$, for $i = 1, 2, \ldots, \text{NNZ}$;
$\text{IROW}(i-1) < \text{IROW}(i)$ or $\text{IROW}(i-1) = \text{IROW}(i)$ and $\text{ICOL}(i-1) < \text{ICOL}(i)$, for $i = 2, 3, \ldots, \text{NNZ}$.

8: OMEGA – ***double precision*** *Input*

On entry: if PRECON = 'S', OMEGA is the relaxation parameter ω to be used in the SSOR method. Otherwise OMEGA need not be initialized and is not referenced.

Constraint: $0.0 < \text{OMEGA} < 2.0$.

9: B(N) – ***double precision*** array *Input*

On entry: the right-hand side vector b.

10: M – INTEGER *Input*

On entry: if METHOD = 'RGMRES', M is the dimension of the restart subspace.

If METHOD = 'BICGSTAB', M is the order ℓ of the polynomial Bi-CGSTAB method.

Otherwise, M is not referenced.

Constraints:

if METHOD = 'RGMRES', $0 < \text{M} \leq \min(\text{N}, 50)$;
if METHOD = 'BICGSTAB', $0 < \text{M} \leq \min(\text{N}, 10)$.

11: TOL – ***double precision*** *Input*

On entry: the required tolerance. Let x_k denote the approximate solution at iteration k, and r_k the corresponding residual. The algorithm is considered to have converged at iteration k if

$$\|r_k\|_\infty \leq \tau \times \left(\|b\|_\infty + \|A\|_\infty \|x_k\|_\infty\right).$$

If $\text{TOL} \leq 0.0$, $\tau = \max\left(\sqrt{\epsilon}, \sqrt{n}\epsilon\right)$ is used, where ϵ is the ***machine precision***. Otherwise $\tau = \max\left(\text{TOL}, 10\epsilon, \sqrt{n}\epsilon\right)$ is used.

Constraint: $\text{TOL} < 1.0$.

12: MAXITN – INTEGER *Input*

On entry: the maximum number of iterations allowed.

Constraint: $\text{MAXITN} \geq 1$.

13: X(N) – ***double precision*** array *Input/Output*

On entry: an initial approximation to the solution vector x.

On exit: an improved approximation to the solution vector x.

14: RNORM – ***double precision*** *Output*

On exit: the final value of the residual norm $\|r_k\|_\infty$, where k is the output value of ITN.

15: ITN – INTEGER *Output*

On exit: the number of iterations carried out.

16: WORK(LWORK) – ***double precision*** array *Workspace*

17: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F11DEF is called.

Constraints:

if METHOD = 'RGMRES', LWORK $\geq 4 \times \mathrm{N} + \mathrm{M} \times (\mathrm{M} + \mathrm{N} + 5) + \nu + 101$;
if METHOD = 'CGS', LWORK $\geq 8 \times \mathrm{N} + \nu + 100$;
if METHOD = 'BICGSTAB', LWORK $\geq 2 \times \mathrm{N} \times (\mathrm{M} + 3) + \mathrm{M} \times (\mathrm{M} + 2) + \nu + 100$;
if METHOD = 'TFQMR', LWORK $\geq 11 \times \mathrm{N} + \nu + 100$.

where $\nu = \mathrm{N}$ for PRECON = 'J' or 'S', and 0 otherwise

18: IWORK($2 \times \mathrm{N} + 1$) – INTEGER array *Workspace*

19: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, METHOD $\neq$ 'RGMRES','CGS','BICGSTAB', or 'TFQMR',
or PRECON $\neq$ 'N', 'J' or 'S',
or $\mathrm{N} < 1$,
or $\mathrm{NNZ} < 1$,
or $\mathrm{NNZ} > \mathrm{N}^2$,
or PRECON = 'S' and OMEGA lies outside the interval $(0.0, 2.0)$,
or $\mathrm{M} < 1$,
or $\mathrm{M} > \min(\mathrm{N}, 50)$, with METHOD = 'RGMRES',
or $\mathrm{M} > \min(\mathrm{N}, 10)$, with METHOD = 'BICGSTAB',
or $\mathrm{TOL} \geq 1.0$,
or $\mathrm{MAXITN} < 1$,
or LWORK too small.

IFAIL = 2

On entry, the arrays IROW and ICOL fail to satisfy the following constraints:

$1 \leq \mathrm{IROW}(i) \leq \mathrm{N}$ and $1 \leq \mathrm{ICOL}(i) \leq \mathrm{N}$, for $i = 1, 2, \ldots, \mathrm{NNZ}$;

$\mathrm{IROW}(i-1) < \mathrm{IROW}(i)$, or $\mathrm{IROW}(i-1) = \mathrm{IROW}(i)$ and $\mathrm{ICOL}(i-1) < \mathrm{ICOL}(i)$, for $i = 2, 3, \ldots, \mathrm{NNZ}$.

Therefore a nonzero element has been supplied which does not lie within the matrix A, is out of order, or has duplicate row and column indices. Call F11ZAF to reorder and sum or remove duplicates.

IFAIL = 3

On entry, the matrix A has a zero diagonal element. Jacobi and SSOR preconditioners are not appropriate for this problem.

IFAIL = 4

The required accuracy could not be obtained. However, a reasonable accuracy may have been obtained, and further iterations could not improve the result. You should check the output value of RNORM for acceptability. This error code usually implies that your problem has been fully and satisfactorily solved to within or close to the accuracy available on your system. Further iterations are unlikely to improve on this situation.

IFAIL = 5

Required accuracy not obtained in MAXITN iterations.

IFAIL = 5 (F11BDF, F11BEF or F11BFF)

A serious error has occurred in an internal call to one of the specified routines. Check all subroutine calls and array sizes. Seek expert help.

7 Accuracy

On successful termination, the final residual $r_k = b - Ax_k$, where $k = \mathrm{ITN}$, satisfies the termination criterion

$$\|r_k\|_\infty \leq \tau \times \left(\|b\|_\infty + \|A\|_\infty \|x_k\|_\infty\right).$$

The value of the final residual norm is returned in RNORM.

8 Further Comments

The time taken by F11DEF for each iteration is roughly proportional to NNZ.

The number of iterations required to achieve a prescribed accuracy cannot be easily determined *a priori*, as it can depend dramatically on the conditioning and spectrum of the preconditioned coefficient matrix $\bar{A} = \mathrm{M}^{-1}A$.

9 Example

This example solves a sparse nonsymmetric system of equations using the RGMRES method, with SSOR preconditioning.

9.1 Program Text

```
*     F11DEF Example Program Text
*     Mark 19 Revised. NAG Copyright 1999.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, LA, LWORK
      PARAMETER        (NMAX=1000,LA=10000,LWORK=10000)
*     .. Local Scalars ..
      DOUBLE PRECISION OMEGA, RNORM, TOL
      INTEGER          I, IFAIL, ITN, L, LWREQ, M, MAXITN, N, NNZ
      CHARACTER        PRECON
      CHARACTER*8      METHOD
*     .. Local Arrays ..
      DOUBLE PRECISION A(LA), B(NMAX), WORK(LWORK), X(NMAX)
      INTEGER          ICOL(LA), IROW(LA), IWORK(2*NMAX+1)
*     .. External Subroutines ..
      EXTERNAL         F11DEF
*     .. Intrinsic Functions ..
```

```
      INTRINSIC        MAX
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F11DEF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
*
*     Read algorithmic parameters
*
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
         READ (NIN,*) NNZ
         READ (NIN,*) METHOD, PRECON
         READ (NIN,*) OMEGA
         READ (NIN,*) M, TOL, MAXITN
*
*        Check size of workspace
*
         L = N
         IF (PRECON.EQ.'N' .OR. PRECON.EQ.'n') L = 0
         LWREQ = MAX(4*N+M*(M+N+5)+L+101,8*N+L+100,2*N*(M+3)+M*(M+2)
     +           +L+100,11*N+L+100)
         IF (LWORK.LT.LWREQ) THEN
            WRITE (NOUT,*) 'LWORK must be at least', LWREQ
            GO TO 60
         END IF
*
*        Read the matrix A
*
         DO 20 I = 1, NNZ
            READ (NIN,*) A(I), IROW(I), ICOL(I)
   20    CONTINUE
*
*        Read right-hand side vector b and initial approximate solution x
*
         READ (NIN,*) (B(I),I=1,N)
         READ (NIN,*) (X(I),I=1,N)
*
*        Solve Ax = b using F11DEF
*
         IFAIL = 1
         CALL F11DEF(METHOD,PRECON,N,NNZ,A,IROW,ICOL,OMEGA,B,M,TOL,
     +               MAXITN,X,RNORM,ITN,WORK,LWORK,IWORK,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,'(A,I10,A)') ' Converged in', ITN, ' iterations'
            WRITE (NOUT,'(A,1P,E16.3)') ' Final residual norm =', RNORM
            WRITE (NOUT,*)
*
*           Output x
*
            WRITE (NOUT,*) '           X'
            DO 40 I = 1, N
               WRITE (NOUT,'(1X,1P,E16.4)') X(I)
   40       CONTINUE
*
         ELSE
            WRITE (NOUT,99999) IFAIL
         END IF
      END IF
   60 CONTINUE
*
99999 FORMAT (1X,/1X,' ** F11DEF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
F11DEF Example Program Data
  5                   N
 16                   NNZ
  'RGMRES' 'S'        METHOD, PRECON
  1.05                 OMEGA
  1 1.D-10 1000       M, TOL, MAXITN
  2.   1    1
  1.   1    2
 -1.   1    4
 -3.   2    2
 -2.   2    3
  1.   2    5
  1.   3    1
  5.   3    3
  3.   3    4
  1.   3    5
 -2.   4    1
 -3.   4    4
 -1.   4    5
  4.   5    2
 -2.   5    3
 -6.   5    5        A(I), IROW(I), ICOL(I), I=1,...,NNZ
  0.  -7.  33.
-19. -28.             B(I), I=1,...,N
  0.   0.   0.
  0.   0.             X(I), I=1,...,N
```

9.3 Program Results

```
 F11DEF Example Program Results

 Converged in        13 iterations
 Final residual norm =       5.087E-09

           X
      1.0000E+00
      2.0000E+00
      3.0000E+00
      4.0000E+00
      5.0000E+00
```